Journal of Software for Algebra and Geometry

Computing with jets

Federico Galetto and Nicholas Iammarino

vol 12 2022

JSAG **12** (2022), 43–49
https://doi.org/10.2140/jsag.2022.12.43

The Journal of Software for
Algebra and Geometry

Computing with jets

FEDERICO GALETTO AND NICHOLAS IAMMARINO

ABSTRACT: We introduce a Macaulay2 package for working with jet schemes. The main method constructs jets of ideals, polynomial rings and their quotients, ring homomorphisms, affine varieties, and (hyper)graphs. The package also includes additional methods to compute principal components and radicals of jets of monomial ideals.

1. INTRODUCTION.

Roughly speaking, the scheme of s-jets of a scheme X is the collection of order s Taylor approximations at points of X. More formally, let X be a scheme over a field $\Bbbk$. Following [Ein and Mustaţă 2009, §2], we call a scheme $\mathcal{J}_s(X)$ over $\Bbbk$ the scheme of s-jets of X, if for every $\Bbbk$-algebra A there is a functorial bijection

$$\operatorname{Hom}(\operatorname{Spec}(A), \mathcal{J}_s(X)) \cong \operatorname{Hom}(\operatorname{Spec}(A[t]/\langle t^{s+1}\rangle), X).$$

This means that the A-points of $\mathcal{J}_s(X)$ are in bijection with the $A[t]/\langle t^{s+1}\rangle$-points of X. It follows that $\mathcal{J}_0(X) \cong X$, and $\mathcal{J}_1(X)$ is the total tangent scheme of X, in line with the definition of tangent space using dual numbers [Hartshorne 1977, II, Exercise 2.8]. Jet schemes play an important role in the study of singularities, as initially suggested by J. Nash [1995], and in connection with other related topics, such as motivic integration and birational geometry [Denef and Loeser 2001; Mustaţă 2001; 2002; Ein and Mustaţă 2009].

The existence of jet schemes is proved in detail in [Ein and Mustaţă 2009, §2]. We recall an essential step, which is the construction of jets of an affine variety. Let X be an affine variety over $\Bbbk$. Consider a closed embedding of X into an affine space $\mathbb{A}^n$ over $\Bbbk$. Let $I = \langle f_1, \dots, f_r\rangle$ be the ideal of $R = \Bbbk[x_1, \dots, x_n]$ corresponding to this embedding. For $s \in \mathbb{N}$, define the polynomial ring

$$\mathcal{J}_s(R) = \Bbbk[x_{i,j} \mid i = 1, \dots, n, \ j = 0, \dots, s].$$

For each $k = 1, \dots, n$, perform the substitution

$$x_k \mapsto x_{k,0} + x_{k,1}t + x_{k,2}t^2 + \cdots + x_{k,s}t^s = \sum_{j=0}^{s} x_{k,j}t^j$$

MSC2020: primary 13-04, 14-04; secondary 05C25, 13F55, 14M12.
Keywords: jets, Macaulay2, monomial ideals, graphs, determinantal varieties.
`Jets` version 1.1

taking elements of R to elements of $\mathcal{J}_s(R)[t]$. This substitution is the "universal s-jet" corresponding to the identity map on $\mathcal{J}_s(X)$ in the functorial bijection above. Applying this substitution to a generator f_i of I gives the decomposition

$$f_i\left(\sum_{j=0}^{s} x_{1,j}t^j, \dots, \sum_{j=0}^{s} x_{n,j}t^j\right) = \sum_{j\geqslant 0} f_{i,j}t^j,$$

where the coefficients $f_{i,j}$ are polynomials in $\mathcal{J}_s(R)$. The *ideal of s-jets* of $I = \langle f_1, \dots, f_r\rangle$ is the ideal of $\mathcal{J}_s(R)$ defined by

$$\mathcal{J}_s(I) = \langle f_{i,j} \mid i = 1, \dots, r,\ j = 0, \dots, s\rangle.$$

The scheme of s-jets of X is $\operatorname{Spec}(\mathcal{J}_s(R)/\mathcal{J}_s(I))$.

This paper introduces the `Jets` package[1] for [Macaulay2], streamlining the process of constructing ideals of jets as indicated above. We adopt the following notation: the variables in the polynomial rings containing the equations of jets have the names of the variables of the original equations with the order of the jets appended to them, and the same subscripts. Moreover, the rings containing the equations of jets are constructed incrementally as towers.

Ideals of jets are computed via the `jets` method applied to objects of type `Ideal`. In addition, the `jets` method can also be applied to objects of type `QuotientRing`, `RingMap`, and `AffineVariety`, with the effects one would expect from applying jet functors. For more information, including grading options, we invite the reader to consult the documentation of the package. Each of the following sections consists of the package being demonstrated in different contexts.

2. Jets of monomial ideals.

As observed in [Goward and Smith 2006], the ideal of jets of a monomial ideal is typically not a monomial ideal.

```
i1 : needsPackage "Jets";

i2 : R=QQ[x,y,z];

i3 : I=ideal(x*y*z);

o3 : Ideal of R

i4 : J2I=jets(2,I);

o4 : Ideal of QQ[x0, y0, z0][x1, y1, z1][x2, y2, z2]

i5 : netList J2I_*

     +-----------------------------------------------------------------+
o5 = |y0*z0*x2 + x0*z0*y2 + x0*y0*z2 + z0*x1*y1 + y0*x1*z1 + x0*y1*z1|
     +-----------------------------------------------------------------+
     |y0*z0*x1 + x0*z0*y1 + x0*y0*z1                                   |
     +-----------------------------------------------------------------+
     |x0*y0*z0                                                         |
     +-----------------------------------------------------------------+
```

However, by [Goward and Smith 2006, Theorem 3.1], the radical is always a (squarefree) monomial ideal. In fact, the proof of [Goward and Smith 2006, Theorem 3.2] shows that the radical is generated by the individual terms of the generators $f_{i,j}$ described in the introduction. This observation provides an alternative algorithm for computing radicals of jets of monomial ideals, which can be faster than the

[1] Available as a supplement to this paper or at https://github.com/galettof/Jets.

default radical computation in Macaulay2.

```
i6 : jetsRadical(2,I);

o6 : Ideal of QQ[x0, y0, z0][x1, y1, z1][x2, y2, z2]

i7 : netList pack(5,oo_*)

     +--------+--------+--------+--------+--------+
o7 = |y0*z0*x2|x0*z0*y2|x0*y0*z2|z0*x1*y1|y0*x1*z1|
     +--------+--------+--------+--------+--------+
     |x0*y1*z1|y0*z0*x1|x0*z0*y1|x0*y0*z1|x0*y0*z0|
     +--------+--------+--------+--------+--------+
```

For a monomial hypersurface, [Goward and Smith 2006, Theorem 3.2] describes the minimal primes of the ideal of jets. Moreover, the main theorem in [Yuen 2006] counts the multiplicity of the jet scheme of a monomial hypersurface along its minimal primes (see also [Yuen 2007b]). We compute the minimal primes, then use Sayfrafi et al.'s `LocalRings` package to compute their multiplicities in the second jet scheme of the example above.

```
i8 : P=minimalPrimes J2I;

i9 : --flatten ring to use LocalRings package
     (A,f)=flattenRing ring J2I;

i10 : needsPackage "LocalRings";

i11 : --quotient by jets ideal as a module
      M=cokernel gens f J2I;

i12 : --compute the multiplicity of the jets along each component
      mult=for p in P list (
          Rp := localRing(A,f p);
          length(M ** Rp)
          );

i13 : netList(pack(4,mingle{P,mult}),HorizontalSpace=>1)

      +--------------------+---+--------------------+---+
o13 = | ideal (z0, y0, x0) | 6 | ideal (z0, y0, z1) | 3 |
      +--------------------+---+--------------------+---+
      | ideal (z0, y0, y1) | 3 | ideal (z0, x0, z1) | 3 |
      +--------------------+---+--------------------+---+
      | ideal (z0, x0, x1) | 3 | ideal (z0, z1, z2) | 1 |
      +--------------------+---+--------------------+---+
      | ideal (y0, x0, y1) | 3 | ideal (y0, x0, x1) | 3 |
      +--------------------+---+--------------------+---+
      | ideal (y0, y1, y2) | 1 | ideal (x0, x1, x2) | 1 |
      +--------------------+---+--------------------+---+
```

3. JETS OF GRAPHS. Jets of graphs were introduced in [Galetto et al. 2021]. Starting with a finite, simple graph G, one may construct a quadratic squarefree monomial ideal $I(G)$ (known as the *edge ideal* of the graph) by converting edges to monomials (see for example [Van Tuyl 2013]). One may then consider the radical of the ideal of s-jets of $I(G)$, which is again a quadratic squarefree monomial ideal. The graph corresponding to this ideal is the graph of s-jets of G, denoted $\mathcal{J}_s(G)$.

Jets of graphs and hypergraphs can be obtained by applying the `jets` method to objects of type `Graph` and `HyperGraph` from the Macaulay2 `EdgeIdeals` package [Francisco et al. 2009] (which is automatically loaded by the `Jets` package). Consider, for example, the graph in Figure 1.

```
i1 : needsPackage "Jets";

i2 : R=QQ[a..e];

i3 : G=graph({{a,c},{a,d},{a,e},{b,c},{b,d},{b,e},{c,e}});
```



Figure 1. The graph G.

We compute the first and second order jets, and list their edges.

```
i4 : J1G=jets(1,G); netList pack(7,edges J1G)

     +--------+--------+--------+--------+--------+--------+--------+
o5 = |{c1, a0}|{d1, a0}|{e1, a0}|{c1, b0}|{d1, b0}|{e1, b0}|{a1, c0}|
     +--------+--------+--------+--------+--------+--------+--------+
     |{b1, c0}|{e1, c0}|{a0, c0}|{b0, c0}|{a1, d0}|{b1, d0}|{a0, d0}|
     +--------+--------+--------+--------+--------+--------+--------+
     |{b0, d0}|{a1, e0}|{b1, e0}|{c1, e0}|{a0, e0}|{b0, e0}|{c0, e0}|
     +--------+--------+--------+--------+--------+--------+--------+

i6 : J2G=jets(2,G); netList pack(7,edges J2G)

     +--------+--------+--------+--------+--------+--------+--------+
o7 = |{a1, c1}|{b1, c1}|{a1, d1}|{b1, d1}|{a1, e1}|{b1, e1}|{c1, e1}|
     +--------+--------+--------+--------+--------+--------+--------+
     |{c2, a0}|{d2, a0}|{e2, a0}|{c1, a0}|{d1, a0}|{e1, a0}|{c2, b0}|
     +--------+--------+--------+--------+--------+--------+--------+
     |{d2, b0}|{e2, b0}|{c1, b0}|{d1, b0}|{e1, b0}|{a2, c0}|{b2, c0}|
     +--------+--------+--------+--------+--------+--------+--------+
     |{e2, c0}|{a1, c0}|{b1, c0}|{e1, c0}|{a0, c0}|{b0, c0}|{a2, d0}|
     +--------+--------+--------+--------+--------+--------+--------+
     |{b2, d0}|{a1, d0}|{b1, d0}|{a0, d0}|{b0, d0}|{a2, e0}|{b2, e0}|
     +--------+--------+--------+--------+--------+--------+--------+
     |{c2, e0}|{a1, e0}|{b1, e0}|{c1, e0}|{a0, e0}|{b0, e0}|{c0, e0}|
     +--------+--------+--------+--------+--------+--------+--------+
```

As predicted in [Galetto et al. 2021, Theorem 3.1], all jets have the same chromatic number.

```
i8 : apply({G,J1G,J2G},chromaticNumber)

o8 = {3, 3, 3}

o8 : List
```

By contrast, jets may not preserve the property of being cochordal.

```
i9 : apply({G,J1G,J2G},x -> isChordal complementGraph x)

o9 = {true, true, false}

o9 : List
```

Using Fröberg’s theorem [1990], we deduce that although the edge ideal of a graph may have a linear free resolution, the edge ideals of its jets may not have linear resolutions.

Finally, we compare minimal vertex covers of the graph and of its second order jets.

```
i10 : vertexCovers G

o10 = {a*b*c, a*b*e, c*d*e}

o10 : List

i11 : netList pack(2,vertexCovers J2G)

      +-------------------------+-------------------------+
o11 = |a2*b2*c2*a1*b1*c1*a0*b0*c0|a2*b2*e2*a1*b1*e1*a0*b0*e0|
      +-------------------------+-------------------------+
      |a2*b2*a1*b1*c1*a0*b0*c0*e0|a2*b2*a1*b1*e1*a0*b0*c0*e0|
      +-------------------------+-------------------------+
      |c2*d2*e2*c1*d1*e1*c0*d0*e0|a1*b1*c1*a0*b0*c0*d0*e0   |
      +-------------------------+-------------------------+
      |a1*b1*e1*a0*b0*c0*d0*e0   |c1*d1*e1*a0*b0*c0*d0*e0   |
      +-------------------------+-------------------------+
```

With the exception of the second row, many vertex covers arise as indicated in [Galetto et al. 2021, Propositions 5.2 and 5.3].

4. Jets of determinantal varieties. Determinantal varieties are classical geometric objects whose jets have been studied with a certain degree of success [Košir and Sethuraman 2005a; 2005b; Yuen 2007a; Ghorpade et al. 2014; Docampo 2013; Mallory 2021]. For our example, we consider the determinantal varieties X_r of 3×3 matrices of rank at most r, which are defined by the vanishing of minors of size $r+1$. We illustrate computationally some of the known results about jets.

```
i1 : needsPackage "Jets";

i2 : R=QQ[x_(1,1)..x_(3,3)];

i3 : G=genericMatrix(R,3,3)

o3 = | x_(1,1) x_(2,1) x_(3,1) |
     | x_(1,2) x_(2,2) x_(3,2) |
     | x_(1,3) x_(2,3) x_(3,3) |

              3       3
o3 : Matrix R  <--- R
```

Since X_0 is a single point, its first jet scheme consists of a single (smooth) point.

```
i4 : I1=minors(1,G);

o4 : Ideal of R

i5 : JI1=jets(1,I1);

o5 : Ideal of QQ[x0   ..x0   ][x1   ..x1   ]
                   1,1    3,3     1,1    3,3

i6 : dim JI1, isPrime JI1

o6 = (0, true)

o6 : Sequence
```

The jets of X_2 (the determinantal hypersurface) are known to be irreducible (see [Košir and Sethuraman 2005a, Theorem 3.1] or [Docampo 2013, Corollary 4.13]). Since X_2 is a complete intersection and has rational singularities [Weyman 2003, Corollary 6.1.5(b)], this also follows from a more general result of M. Mustaţă [2001, Theorem 3.3].

```
i7 : I3=minors(3,G);

o7 : Ideal of R

i8 : JI3=jets(1,I3);

o8 : Ideal of QQ[x0   ..x0   ][x1   ..x1   ]
                   1,1    3,3     1,1    3,3

i9 : isPrime JI3

o9 = true
```

For the case of 2×2 minors, [Košir and Sethuraman 2005a, Theorem 5.1], [Yuen 2007a, Theorem 5.1], and [Docampo 2013, Corollary 4.13] all count the number of components; the first two of these references describe the components further. As expected, the first jet scheme of X_1 has two components, one of them an affine space.

```
i10 : I2=minors(2,G);

o10 : Ideal of R
```

```
i11 : JI2=jets(1,I2);
o11 : Ideal of QQ[x0   ..x0   ][x1   ..x1   ]
                    1,1    3,3    1,1    3,3
i12 : P=primaryDecomposition JI2; #P
o13 = 2
i14 : P_1
o14 = ideal (x0   , x0   , x0   , x0   , x0   , x0   , x0   , x0   , x0   )
               3,3    3,2    3,1    2,3    2,2    2,1    1,3    1,2    1,1
o14 : Ideal of QQ[x0   ..x0   ][x1   ..x1   ]
                    1,1    3,3    1,1    3,3
```

The other component is the so-called principal component of the jet scheme, i.e., the Zariski closure of the first jets of the smooth locus of X_1. To check this, we first establish that the first jet scheme is reduced (i.e., its ideal is radical), then use the `principalComponent` method with the option `Saturate=>false` to speed up computations. (We invite the reader to consult the package documentation for more details.)

```
i15 : radical JI2==JI2
o15 = true
i16 : P_0 == principalComponent(1,I2,Saturate=>false)
o16 = true
```

Finally, as observed in [Ghorpade et al. 2014, Theorem 18], the Hilbert series of the principal component of the first jet scheme of X_1 is the square of the Hilbert series of X_1.

```
i17 : apply({P_0,I2}, X -> hilbertSeries(X,Reduce=>true))
                     2      3    4             2
        1 + 8T + 18T  + 8T  + T    1 + 4T + T
o17 = {------------------------, -----------}
                       10                 5
                (1 - T)            (1 - T)
o17 : List
i18 : numerator (first oo) == (numerator last oo)^2
o18 = true
```

SUPPLEMENT. The online supplement contains version 1.1 of `Jets`.

ACKNOWLEDGEMENTS. We are grateful to Greg Smith for valuable feedback on an early version of the package. We also thank the anonymous referees for suggesting several improvements to the original version of this article and package.

REFERENCES.

[Denef and Loeser 2001] J. Denef and F. Loeser, "Geometry on arc spaces of algebraic varieties", pp. 327–348 in *European Congress of Mathematics*, *Vol. I* (Barcelona, 2000), edited by C. Casacuberta et al., Progr. Math. **201**, Birkhäuser, Basel, 2001. MR

[Docampo 2013] R. Docampo, "Arcs on determinantal varieties", *Trans. Amer. Math. Soc.* **365**:5 (2013), 2241–2269. MR Zbl

[Ein and Mustaţă 2009] L. Ein and M. Mustaţă, "Jet schemes and singularities", pp. 505–546 in *Algebraic geometry* (Seattle, 2005), edited by D. Abramovich et al., Proc. Sympos. Pure Math. **80**, Amer. Math. Soc., Providence, RI, 2009. MR Zbl

[Francisco et al. 2009] C. A. Francisco, A. Hoefel, and A. Van Tuyl, "EdgeIdeals: a package for (hyper)graphs", *J. Softw. Algebra Geom.* **1** (2009), 1–4. MR Zbl

[Fröberg 1990] R. Fröberg, "On Stanley–Reisner rings", pp. 57–70 in *Topics in algebra* (Warsaw, 1988), edited by S. Balcerzyk et al., Banach Center Publ. **26**, PWN, Warsaw, 1990. MR Zbl

[Galetto et al. 2021] F. Galetto, E. Helmick, and M. Walsh, "Jet graphs", *Involve* **14**:5 (2021), 793–812. MR Zbl

[Ghorpade et al. 2014] S. R. Ghorpade, B. Jonov, and B. A. Sethuraman, "Hilbert series of certain jet schemes of determinantal varieties", *Pacific J. Math.* **272**:1 (2014), 147–175. MR Zbl

[Goward and Smith 2006] R. A. Goward, Jr. and K. E. Smith, "The jet scheme of a monomial scheme", *Comm. Algebra* **34**:5 (2006), 1591–1598. MR Zbl

[Hartshorne 1977] R. Hartshorne, *Algebraic geometry*, Graduate Texts in Mathematics **52**, Springer, 1977. MR Zbl

[Košir and Sethuraman 2005a] T. Košir and B. A. Sethuraman, "Determinantal varieties over truncated polynomial rings", *J. Pure Appl. Algebra* **195**:1 (2005), 75–95. MR Zbl

[Košir and Sethuraman 2005b] T. Košir and B. A. Sethuraman, "A Groebner basis for the 2×2 determinantal ideal $\mod t^2$", *J. Algebra* **292**:1 (2005), 138–153. MR

[Macaulay2] D. R. Grayson and M. E. Stillman, "Macaulay2: a software system for research in algebraic geometry", available at http://www.math.uiuc.edu/Macaulay2.

[Mallory 2021] D. Mallory, "Minimal log discrepancies of determinantal varieties via jet schemes", *J. Pure Appl. Algebra* **225**:2 (2021), art. id. 106497. MR Zbl

[Mustaţă 2002] M. Mustaţă, "Singularities of pairs via jet schemes", *J. Amer. Math. Soc.* **15**:3 (2002), 599–615. MR Zbl

[Mustaţă 2001] M. Mustaţă, "Jet schemes of locally complete intersection canonical singularities", *Invent. Math.* **145**:3 (2001), 397–424. MR Zbl

[Nash 1995] J. F. Nash, Jr., "Arc structure of singularities", *Duke Math. J.* **81**:1 (1995), 31–38. MR Zbl

[Van Tuyl 2013] A. Van Tuyl, "A beginner's guide to edge and cover ideals", pp. 63–94 in *Monomial ideals, computations and applications*, edited by P. G. Anna M. Bigatti and E. S. de Cabezón, Lecture Notes in Math. **2083**, Springer, 2013. MR Zbl

[Weyman 2003] J. Weyman, *Cohomology of vector bundles and syzygies*, Cambridge Tracts in Mathematics **149**, Cambridge University Press, 2003. MR Zbl

[Yuen 2006] C. Yuen, "Multiplicity of jet schemes of monomial schemes", 2006. arXiv math/0607638

[Yuen 2007a] C. Yuen, "Jet schemes of determinantal varieties", pp. 261–270 in *Algebra, geometry and their interactions*, edited by J. M. Alberto Corso and C. Polini, Contemp. Math. **448**, Amer. Math. Soc., Providence, RI, 2007. MR Zbl

[Yuen 2007b] C. Yuen, "The multiplicity of jet schemes of a simple normal crossing divisor", *Comm. Algebra* **35**:12 (2007), 3909–3911. MR Zbl

RECEIVED: 12 Aug 2021 REVISED: 13 Jun 2022 ACCEPTED: 20 Oct 2022

FEDERICO GALETTO:

f.galetto@csuohio.edu
Department of Mathematics and Statistics, Cleveland State University, Cleveland, OH, United States

NICHOLAS IAMMARINO:

nickiammarino@gmail.com
Department of Mathematics and Statistics, Cleveland State University, Cleveland, OH, United States